ORF 307: Lecture 19

Linear Programming: Chapter 13, Section 2
Pricing American Options

Robert Vanderbei

May 1, 2018

Slides last edited on April 30, 2018
A *Perpetual American Option* is a legal contract giving the holder of the option the right to buy a particular stock at a particular price, say $K$, at any time in the future.

This price is called the *strike price*.

Let $X_n$ denote the share price of the stock at time $n$ (i.e., $n$ days in the future).

Obviously, $X_n$ only becomes known at time $n$. Prior to time $n$ it is a random variable.

To keep things as simple as possible, we assume that at the end of each day the share price can do only one of two things:

- Go up by a fixed small factor, or
- Go down by same fixed small factor.

So, if the price today is $X_0$, the price tomorrow will be either $X_0 \times r$ or $X_0/r$, where $r$ is a specific factor close to, but slightly larger than, one, say, for example, $r = 1.008$. 

Because our simple model allows only two choices for how the share price changes from one day to the next, it follows that after \( n \) days there are only a fixed (finite) collection of possible share prices.

To see it, suppose that up to day \( n \) there has been \( j \) up days and \( k \) down days. Then, clearly, the share price on day \( n \) is

\[
X_n = X_0 \times r^{j-k}.
\]

Hence, the set of states that the share price can be “in” is simply

\[
\ldots, X_0r^{-3}, X_0r^{-2}, X_0r^{-1}, X_0, X_0r, X_0r^2, X_0r^3, \ldots
\]
Suppose that, on day $n$, the holder of the option decides to “exercise” the option.

If the share price $X_n$ is greater than the strike price $K$, then the holder of the option can buy the stock for $K$ dollars and immediately sell it for $X_n$ dollars and realize a gain of $X_n - K$ dollars.

If, on the other hand, the share price is less than the strike price, the option holder would lose money if he/she were to exercise the option.

The value of the option if exercised on day $n$ is

$$f(X_n) = X_n - K.$$
We suppose the option holder chooses to exercise the option at some future time $\tau$, the choice of which can depend on the evolution of the share price between now and then. That is, on the first day ($n = 0$), $\tau$ must be modeled as a random variable.

At $n = 0$, we want to know the expected value of the payoff:

$$E \alpha^\tau f(X_\tau).$$

Here, since the option is perpetual, it is important to introduce a discount rate $\alpha$. It is a number slightly less than one. It represents today’s value of tomorrow’s dollar.

The optimal strategy is determined by maximizing over all (non-clairvoyant) random times $\tau$:

$$v(x) = \max_\tau E (\alpha^\tau f(X_\tau) \mid X_0 = x)$$
Suppose the market has been evolving for a while and we still have the option. Suppose the current share price is $x$.

Let’s consider our choices. We can either decide to exercise the option or hold it until tomorrow at which time the same question will be asked again.

If we exercise the option, then we get $f(x)$ dollars.

If we hold until tomorrow (or beyond) and assume that we will behave optimally from then on, then the share price either goes up by a factor of $r$ or down by a factor of $r$ with equal probability. Hence, the value of this choice is $(v(xr) + v(x/r))/2$. But, that’s in tomorrow’s dollars. To convert to today’s dollars, we have to multiply by the discount factor $\alpha$.

Obviously we should pick the better of the two options:

$$v(x) = \max \left( f(x), \alpha \frac{v(xr) + v(x/r)}{2} \right).$$
Converting to an LP Problem

One can show that $v(x)$ is the smallest function that satisfies:

\[
\begin{align*}
v(x) & \geq f(x), \\
v(x) & \geq \alpha \frac{v(rx) + v(x/r)}{2}
\end{align*}
\]

Since the set of $x$’s is discrete, there are really only a discrete set of unknowns $v(x)$, $x = \ldots, X_0r^{-3}, X_0r^{-2}, X_0r^{-1}, X_0, X_0r, X_0r^2, X_0r^3, \ldots$.

If we arbitrarily truncate that set, then we get the following linear programming problem:

\[
\begin{align*}
\text{minimize} & \quad \sum_x v(x) \\
\text{subject to} & \quad v(x) \geq f(x) \quad \text{for all } x \\
& \quad v(x) \geq \alpha \frac{v(rx) + v(x/r)}{2} \quad \text{for all } x \\
& \quad v(x) \geq 0 \quad \text{for all } x.
\end{align*}
\]
reset;

param x0 := 110;
param K := 100;
param r := 1.008;
param p := 0.5;
param q := 1-p;
param alpha := 1/1.55^(1/360);

check: 1/r <= alpha;
check: alpha <= r;

param N := 1000;

set X := setof {j in -N..N: x0*r^j <= 250 && x0*r^j >= 1} x0*r^j ordered;

var v {X} >= 0;

minimize vx0: sum {x in X} v[x];

subject to waitOneTimeUnit {x in X: x != first(X) && x != last(X)}:
   v[x] >= alpha*(p*v[next(x)] + q*v[prev(x)]);

subject to exerciseNow{x in X}: v[x] >= x-K;

solve;

printf {x in X}: "%8.3f %8.3f %8.3f \n", x, max(x-K, 0), v[x];
printf "%8.3f %8.3f \n", x0, v[x0];
display 1/alpha^360;
display 1/r, alpha, p*r+q/r, 1/alpha, r;
Exercise option if and only if stock price exceeds $204.